Two problems of Jakob Bernoulli

Richard J.Pulskamp
27 November 2002

Jakob Bernoulli posed two problems in the *Journal of Sçavans* of 1685. No solutions were received. His own solution was published in the *Acta Eruditorum* of 1690 and, in the issue for June of the same journal, that of Leibnitz.

Problem 1. A & B play with a die, on the condition that the one who casts first will have won. A plays one time, then B one time; after A plays twice in sequence, then B twice; then A three times in sequence, & B also three times.

What he means by the condition is that the one who casts the die and is the first to achieve a certain event, such as casting a 1, wins.

Solution: To find the probability that A win it is necessary to sum an infinite series which does not have a closed form.

The sequence of casts will be ABAABBAAABBBAAAABBBB...

Note that A appears in the positions 1, 3,4, 7,8,9 13,14,15,16, 21,22,23,24,25, ... The last position of each subsequence of length n is the corresponding square n^2.

In the same manner, B appears in the positions 2, 5,6, 10,11,12, 17,18,19,20, 26,27,28,29,30, ... The last position of each subsequence of length n is the product of two consecutive integers $n\,(n+1)$.

It is therefore easy to write down the probability that each win as a double sum.

Let p denote the probability that a player achieves the event on a given cast and let $q = 1 - p$. The probability that the casting player win on the kth trial is $p\,q^{(k-1)}$. Finally, let P and Q denote the respective probabilities that A and B win.

```
> q:=1-p;
```

$$q := 1 - p$$

```
> P:=sum(sum(p*q^(k-1),k=n^2-n+1..n^2),n=1..infinity);
```

$$P := \sum_{n=1}^{\infty} \left(\frac{(1-p)^{(n^2+1)}}{-1+p} - \frac{(1-p)^{(n^2-n+1)}}{-1+p} \right)$$

```
> Q:=sum(sum(p*q^(k-1),k=n^2+1..n*(n+1)),n=1..infinity);
```

$$Q := \sum_{n=1}^{\infty} \left(\frac{(1-p)^{(n(n+1)+1)}}{-1+p} - \frac{(1-p)^{(n^2+1)}}{-1+p} \right)$$

Bernoulli takes the event to be casting a 7 with two dice. This gives p=1/6. His approximate solution produces the odds of 59679 to 40321. The computations below confirm his result.

```
> subs(p=1/6,P):
> R:=evalf(%);
```

$$R := 0.5967919434$$

```
> subs(p=1/6,Q):
> S:=evalf(%);
```

$$S := 0.4032080566$$

```
>
```

Problem 2. A plays one time, then B two times in sequence, then A three times in sequence, then B four times, &c. until one of them wins.

Solution: The sequence of casts will be ABBAAABBBBAAAAABBBBBB...

Note that A appears in positions 1, 4,5,6, 11,12,13,14,15, 22,23,24,25,26,27,28, ... The sequence of length n ends at position $\frac{n(n+1)}{2}$. These sequences are all of odd length.

Note that B appears in positions 2,3, 7,8,9,10, 16,17,18,19,20,21, 29,30,31,32,33,34,35,36, ...

The sequence of length n ends at position $\frac{n(n+1)}{2}$ as well. Each sequence is even in length.

Bernoulli obtains the approximate odds of 52392 to 47608. This is confirmed below.

```
> P2:=sum(sum(p*q^(k-1),k=2*n^2-3*n+2..(2*n-1)*n),n=1..infinity);
```

$$P2 := \sum_{n=1}^{\infty} \left(\frac{(1-p)^{((2n-1)n+1)}}{-1+p} - \frac{(1-p)^{(2n^2-3n+2)}}{-1+p} \right)$$

```
> Q2:=sum(sum(p*q^(k-1),k=2*n^2-n+1..n*(2*n+1)),n=1..infinity);
```

$$Q2 := \sum_{n=1}^{\infty} \left(\frac{(1-p)^{(n(2n+1)+1)}}{-1+p} - \frac{(1-p)^{(2n^2-n+1)}}{-1+p} \right)$$

```
> subs(p=1/6,P2):
> evalf(%);
```

$$0.5239191276$$

```
> subs(p=1/6,Q2):
> evalf(%);
```

$$0.4760808724$$

Bernoulli's solution as published in Acta Eruditorum 1690.

```
> sum((5/6)^(n*(n+1)),n=0..infinity)-sum((5/6)^(n^2),n=1..infinity);
```

$$\left(\sum_{n=0}^{\infty}\left(\frac{5}{6}\right)^{(n\,(n+1))}\right)-\left(\sum_{n=1}^{\infty}\left(\frac{5}{6}\right)^{(n^2)}\right)$$

```
> evalf(%);
```

$$0.596791943$$

```
> sum((5/6)^(2*n*(2*n+1)/2),n=0..infinity)-sum((5/6)^((2*n+1)*(n+1))
,n=0..infinity);
```

$$\left(\sum_{n=0}^{\infty}\left(\frac{5}{6}\right)^{(n\,(2\,n+1))}\right)-\left(\sum_{n=0}^{\infty}\left(\frac{5}{6}\right)^{((2\,n+1)\,(n+1))}\right)$$

```
> evalf(%);
```

$$0.523919128$$

```
>
```